Creating Metadata-Driven Program Logic with the SAS Call Execute Routine - An Example with CTC Grading of Laboratory Results

Robert W. Graebner, Quintiles, Inc., Overland Park, KS

ABSTRACT
Certain programming tasks require the application of complex, data-dependent rules in the derivation of classification variables. One such example is assigning a CTC grade to laboratory results. Determination of the CTC grade depends on the laboratory parameter and its value in relation to either fixed values or multiples of the upper and lower limits of the normal range. This paper presents a solution in which a data step is used to access CTC grading logic metadata and the CALL EXECUTE routine is used to place the associated SAS statements in the SAS command stack for execution after the data step has finished. This process creates a virtual data step that will assign the CTC grades to each observation in the laboratory results data set.

INTRODUCTION
Applications that utilize complex logic can require large, nested blocks of conditional statements. These statements may be based on the values of several variables, and the inclusion of some variables in the process may be conditional. CTC grading of lab results can require 9 to 11 logic statements per lab parameter. This typically results in large volumes of tedious source code that can be difficult to debug and maintain. If you add the possibility that the criteria for the decision rules may change over time, you quickly find yourself facing a maintenance nightmare if you adopt traditional techniques.

One approach to managing such a process is to build a “Logic Processing” procedure that bases the logic rules that are applied to the lab data on another set of data that define the rules. The data that define the logic process to be performed can be referred to as metadata. Creating such a data set can be straightforward, however writing a program that can access this data and then form the necessary conditional statements to apply the rules can be difficult. Fortunately, the SAS CALL EXECUTE routine can greatly simplify the task.

Before getting into the details of how the CALL EXECUTE routine can be used for this task, let us take a look at a simple example to get a feel for how the basic process works. A common first task used when learning a new programming language is to print the text string “Hello World!”. Using this task as our starting point, we can create a short program that uses CALL EXECUTE in a simple, if indirect way of printing “Hello World!” in the SAS log. The program consists of a data step with three CALL EXECUTE commands. Each of these commands contains a single argument that is a text string containing one of the statements in a data step used to actually print “Hello World!” in the SAS log. The source code for the program is shown below.

```sas
data test;
  call execute("Data _null_;");
  call execute("put 'Hello World!';");
  call execute("run;");
run;
```
When we run the program, we get the following SAS log:

```
data test;
call execute("Data _null_;");
call execute("put 'Hello World!';");
call execute("run;");
run;
```

NOTE: The data set WORK.TEST has 1 observations and 0 variables.
NOTE: DATA statement used (Total process time):
real time 0.77 seconds
cpu time 0.11 seconds

When the data step runs, each of the CALL EXECUTE statements sends the enclosed text string to the SAS macro processor and the resulting string is placed in a command stack. In this simple example, each argument consists entirely of verbatim text. The real power of this approach is achieved when fixed text is combined with the contents of data step or macro variables to form executable SAS statements. When the data step has completed, SAS begins processing the commands placed in the stack prior to processing any statements that follow the data step. The SAS log provides a note regarding the generated lines and below that we see the result of the PUT statement: "Hello World!". This is a trivial example, but the basic process can be used to generate SAS code based on metadata and then run it immediately. In this paper, I use the process of CTC grading to illustrate a more complex example of the use of metadata in logic processing. Therefore, before proceeding to a more detailed look at the CALL EXECUTE routine, I will give a brief overview of the CTC grading process.

**CTC GRADING – AN OVERVIEW**

The National Cancer Institute Common Toxicity Criteria (CTC) was developed as a method of grading the severity of adverse events. CTC grading uses an ordinal scale ranging from 0 to 5, representing none or normal, mild, moderate, severe, life threatening or disabling, and fatal. While these general terms are associated with each grade, within individual categories, there are specific clinical criteria associated with each grade. The example in this paper is based on laboratory results within the Blood / Bone Marrow and Metabolic / Laboratory categories. More detailed information on CTC grading can be found at [http://ctep.cancer.gov/forms/CTCAEv3.pdf](http://ctep.cancer.gov/forms/CTCAEv3.pdf).

CTC grading of laboratory results indicates how far above or below normal a value lies, expressed in terms of the potential impact of such a value on the subject’s health and well being. For the example in this paper, CTC grades can range from –4 to 4 for each lab parameter result. A grade of 0 indicates that the result is within the normal range. Positive grades represent values above the normal range (hyper) and negative grades represent values below normal (hypo). In the formal definition of the CTC grades, the hypo- and hyper- conditions would be specified separately, with a range of 0 to 4 for each. Combining both conditions simplifies the program logic associated with assigning CTC grades. Each grade level has associated with it an upper and/or lower limit. These limits can be specified as either an absolute value or as a multiplier to be applied to the upper or lower normal range limit. The logical operator used to compare the laboratory result to each limit is also specified. An example of a SAS data set containing CTC grading metadata is given in Figure 1.
The logic rules associated with assigning CTC grades can be specified in a SAS data set containing as few as eight variables. An observation can be created for each laboratory parameter and CTC grade combination, resulting in a process logic metadata table. An example of such a data set is given below.

Assuming that units have been standardized, variables are needed for parameter name (PARNAM1C), CTC grade, lower and upper grade limits (LOW_VAL and UPP_VAL), comparison operators (LOW_COMP and UPP_COMP), and limit reference (LOW_REF and UPP_REF). If the limit reference is "LNR" or "UNR", then the corresponding range limit is obtained by multiplying the value in LOW_VAL or UPP_VAL by the upper or lower normal range limit from the laboratory result data set. If the limit reference is blank, the value in LOW_VAL or UPP_VAL is used as the limit. The next step is to combine the logic metadata in this data set with the laboratory results data set to facilitate the CTC grading process.

Figure 1. SAS data set containing CTC grading metadata

ASSESSING CONVENTIONAL APPROACHES

Simpler processing tasks can be accomplished with more traditional approaches. For example, applying simple normal ranges to laboratory results can be accomplished by merging the two data sets and performing simple comparisons. More complex examples include applying normal ranges that are dependent on the subject’s sex, age or weight or on the date of analysis. In this case, some of the identifier variables must match exactly, others (e.g. age or date) must fall within the range specified in the normal range file. This can be accomplished with a data step merge using the exact matching variables in the by statement, and the using a WHERE or IF statement to keep only those observations that meet the criteria. Another approach is to perform a fuzzy merge using a PROC SQL SELECT statement with appropriate GROUP BY and HAVING clauses.

With the example of determining CTC grading of laboratory results, the problem becomes more complex. The grading of each laboratory result is dependent on multiple rows in the CTC logic metadata. Not only are there multiple comparison ranges for each parameter, but some of the upper and lower limits are absolute values while others specify a multiple of the upper or lower normal range limit. To further complicate the process, the specific comparison operator (<, <=, >, >=) can vary as well. The grading process could be accomplished using a SELECT / WHEN block that contained a WHEN statement for each parameter and grade combination. However this approach would be difficult to maintain, requiring the program to be modified for any changes to the CTC grading criteria or for the addition of new laboratory parameters. Another possibility, thanks to the power and flexibility of the SAS CALL EXECUTE routine, is to read in logic metadata which specify the necessary comparison parameters for each CTC grade and build a data step that will apply them to the laboratory results.
THE SAS CALL EXECUTE ROUTINE

Before delving into the specifics of generating a data step, it is important to have a basic understanding of the SAS CALL EXECUTE routine. The SAS CALL EXECUTE routine has a single argument that may contain literal text strings, data step variable names or SAS macro statements. When the CALL EXECUTE routine is executed, the processing of the current data step is temporarily stopped and the text string provided as the argument is sent to the macro processor to be resolved. Any SAS macro variable references are resolved and then any macro statements, functions or calls are processed. The text string produced by the macro processor is then placed in the command stack. When the current data step has completed, SAS begins processing the lines of text in the command stack and attempts to interpret them as valid SAS program statements. When all of the lines in the command stack have been processed, SAS begins executing any statements that follow the data step.

When macro statements are included in the argument to CALL EXECUTE, it is crucial to consider the order of events and the flow of control between the SAS macro processor and the data step. It is also important to remember that the macro processor will attempt to resolve text contained within double quotes, while text in single quotes will be treated as literal text and will be simply passed through the macro processor. One example where double quotes are necessary is when you include a macro variable reference in a text string and you want the macro variable to be resolved as part of creating the SAS statement to be sent to the command stack. Single quotes are necessary if you want the macro variable reference to be resolved when the SAS statement you create is actually executed.

GENERATING A DATA STEP FROM METADATA

A key challenge associated with basing data step programming logic on metadata stored in a data set is timing. We need to access variables in the logic metadata to build the necessary program statements and then have those statements available to process laboratory data from another data set. This process requires the use of two data steps; one to loop through the logic metadata and generate valid SAS statements, and a second to process the laboratory results data set using the logic statements generated by the first. This may seem like a daunting task, however it can be accomplished using the SAS CALL EXECUTE routine. An overview of the process is illustrated below.

By sending a series of commands that form a data step to the command stack, we can use one data step to create another. Since the second data step is created by the first, they in essence form an intertwined data step pair, consisting of a real, constructor data step, and a virtual, logic processing data step. In the CTC grading example, a virtual SAS data step with a SELECT / WHEN block to assign a CTC grade to each observation of laboratory data is constructed line by line and placed in the SAS command stack. The second data step is completely dependent on the first for its properties, but completely independent as far as processing goes. This basic process is illustrated in Figure 2.

---

**Figure 2. The process of creating a CTC grading data step from metadata**
ANATOMY OF A CONSTRUCTOR DATA STEP

The following example illustrates the basic structure of a constructor data step. First, the CTC grading logic metadata data set and end-of-file flag are specified in a SET statement. Then the text variable that will be use to construct lines of SAS source code is defined. In this example, a DATA _NULL_ step is used, however a temporary or permanent SAS data set name can be specified in order to save the source code lines you generate. This can be particularly helpful during debugging.

```
data _null_;  
set CTC_GRAD end = eof;  
length CODELINE $200;
```

Next, since the processor data step statements in the command stack will be executed immediately after the constructor data step is finished, it is important to test the values in each record of metadata to ensure that they will create a valid line of SAS code. In the example below, only observations with the minimum sufficient subset of metadata parameters are retained. A subsetting IF statement will allow the current iteration of the data step to continue if the parameter name, grade, upper value and lower value are present. Otherwise a warning is printed in the SAS log and the data step skips to the next observation in the metadata data set.

```
if PARNAME ^='' and GRADE ^= . and (LOW_VAL ^= . or UPP_VAL ^= .));
else put '**** WARNING: Incomplete Metadata Observation ****';
```

Prior to generating SAS code based on the first observation of metadata, the opening statements of the processor data step and the beginning of the SELECT block are placed in the command stack by using CALL EXECUTE with literal text arguments.

```
if _n_= 1 then
do;
call execute('data LABCTC;');
call execute(' set LAB;');
call execute(' select;');
end;
```

The following SAS code is used to construct the individual WHEN statements within the SELECT / WHEN block. Each valid observation in the CTC process logic data set will result in a WHEN statement that will perform the logic test for a single parameter and grade combination.
In the block of SAS code below, a character variable which contains source code for the lower limit of the CTC grade (LOW_LIM) is created, based on the lower limits of the current parameter and grade (LOW_VAL) and the value of the limit reference variable (LOW_REF). If the reference variable is missing, then the grade limit variable’s value is used as an absolute value. If the reference variable is equal to ‘LNR’ or ‘UNR’, then the grade limit value is multiplied by the lower (LABLLN) or upper (LABULN) normal range value to determine the limits to be used for the grade. A similar block of code is used to create the variable UPP_LIM for the upper limit, based on UPP_REF and UPP_VAL.

```sas
if LOW_VAL ^= . then
    do;
    select( upcase( LOW_REF ));
    when('LNR') low_lim = put(LOW_VAL, best12.) || '* LABLLN';
    when('UNR') low_lim = put(LOW_VAL, best12.) || '* LABULN';
    when('')    low_lim = put(LOW_VAL, best12.);
    otherwise low_lim = '';
    end;
end;
else low_lim = ''; 

if upp_lim ^= '' or low_lim ^= '' then
    do;
    codeline = '    when(PARNAME="'||strip(PARNAME)||'" and UNITS="'||strip(UNITS)||'" and ';
        if LOW_VAL ^= . and LOW_COMP ^= '' then
            codeline = strip(codeline) || ' LABRES ' || strip(LOW_COMP) ||
            strip(low_lim));
        if LOW_VAL ^= . and LOW_COMP ^= '' and UPP_VAL ^= . and UPP_COMP ^= '' then
            codeline = strip(codeline) || ' and ';
        if UPP_VAL ^= . and UPP_COMP ^= '' then
            codeline = strip(codeline) || ' LABRES ' || strip(UPP_COMP) ||
            strip(upp_lim);
        if low_ref = 'LNR' or upp_ref = 'LNR' then
            codeline = strip(codeline) || ' and LABLLN ^= . '; 
        if low_ref = 'UNR' or upp_ref = 'UNR' then
            codeline = strip(codeline) || ' and LABULN ^= . '; 
        codeline = strip(codeline) || ' and LABRES ^= . ) CTC_1C = ' ||
            strip( put( GRADE, best12. )) || ';';
    call execute( codeline );
    end;
```

The SAS code in the second block below is used to construct a WHEN statement that handles the logic for the laboratory parameter and CTC grade specified in the current observation of the CTC grading metadata. The WHEN statement is created by combining literal text strings and the contents of metadata variables into a text variable called CODELINE. When the statement is complete, the CALL EXECUTE routine is used to place the contents of CODELINE in the command stack.
When the last line of SAS code that is dependent on metadata has been generated, the final lines of the processor data step are added to the command stack.

```sas
if eof then
do;
call execute('    otherwise;');
call execute('  end;');
call execute('run;');
end;
run;
```

**ANATOMY OF A PROCESSOR DATA STEP**

The processor data step is relatively simple, but with many parts. The data step contains a single SELECT / WHEN block with a WHEN statement for each CTC grade that can be associated with each laboratory parameter. The good news is that you did not have to type it all in, or find the correct location to make changes if the criteria change. A small subsection of a processor data step is shown below.

```sas
data lrs5ctc;
  set lrs5;
  select;
  when(PARNAM="CALC" and LABRES >2.4 and LABRES <=2.6) CTCGRD = 0;
  when(PARNAM="CALC" and LABRES >2.6 and LABRES <=2.9) CTCGRD = 1;
  when(PARNAM="CALC" and LABRES >2.9 and LABRES <=3.1) CTCGRD = 2;
  when(PARNAM="CALC" and LABRES >3.1 and LABRES <=3.4) CTCGRD = 3;
  when(PARNAM="CALC" and LABRES >3.4) CTCGRD = 4;
  ... 
  ... 
  otherwise;
  end;
run;
```
CONCLUSION
The example in this paper illustrates how a complex set of logic rules can be defined in a data set and then applied in a relatively simple program. This approach can significantly reduce programming development and maintenance costs, and it can facilitate the maintenance of logic rule sets by nonprogrammers. Once the basics of this method are understood, programming complex logic rules can be a simple task. If SAS macro statements or variables are used in the argument for CALL EXECUTE, consideration of process timing issues is critical. One potential change to increase the efficiency of the processor data step would be to generate nested SELECT / WHEN blocks, using an outer set of WHEN statements to test for laboratory parameter, and inner SELECT / WHEN blocks to test for value ranges. This could be accomplished by specifying the laboratory parameter as a BY-variable in the constructor data step and then using logic statements based on FIRST.PARNAME and LAST.PARNAME.

CONTACT INFORMATION
Robert Graebner
Quintiles, Inc.
P.O. Box 7913
Overland Park, KS  66207

Email: bob.graebner@quintiles.com
       mgraebner@kc.rr.com

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. @ indicates USA registration.